
11 Policy Gradient Estimation

The previous chapter discussed several ways to go about directly optimizing
the parameters of a policy to maximize expected utility. In many applications,
it is often useful to use the gradient of the utility with respect to the policy
parameters to guide the optimization process. This chapter discusses several
approaches to estimating this gradient from trajectory rollouts. A major challenge
with this approach is the variance of the estimate due to the stochastic nature of
the trajectories arising from both the environment and our exploration of it. The
next chapter will discuss how to use these algorithms for gradient estimation for
the purpose of policy optimization.

11.1 Finite Difference

Finite difference methods estimate the gradient of a function from small changes in
its evaluation. Recall that the derivative of a univariate function f is

d f
dx

(x) = lim
δ→0

f (x + δ)− f (x)
δ

(11.1)

The derivative at x can be approximated by a sufficiently small step δ > 0:

d f
dx

(x) ≈ f (x + δ)− f (x)
δ

(11.2)

This approximation is illustrated in figure 11.1.

f (x)

x

Figure 11.1. The finite difference
method approximates the deriva-
tive of f (x) using an evaluation of
a point near x. The finite difference
approximation in red is not a per-
fect match to the true derivative,
drawn in blue.

The gradient of a multivariate function f with an input of length n is:

∇ f (x) =
[

∂ f
∂x1

(x), . . . ,
∂ f
∂xn

(x)
]

(11.3)

Finite differences can be applied to each dimension to estimate the gradient.

In the context of policy optimization, we want to estimate the gradient of the
utility expected from following a policy parameterized by θ:

∇U(θ) =

[
∂U
∂θ1

(θ), . . . ,
∂U
∂θn

(θ)

]
(11.4)

≈
[

U(θ+ δe(1))−U(θ)

δ
, . . . ,

U(θ+ δe(n))−U(θ)

δ

]
(11.5)

where e(i) is the ith standard basis vector consisting of zeros except for the ith
component that is set to 1.

As discussed in section 10.1, we need to simulate policy rollouts to estimate
U(θ). We can use algorithm 11.1 to generate trajectories. From these trajectories,
we can compute their return and estimate the utility associated with the policy.
Algorithm 11.2 implements the gradient estimate in equation (11.5) by simulating
m rollouts for each component and averaging the returns.

function simulate(𝒫::MDP, s, π, d)

τ = []

for i = 1:d

a = π(s)

s′, r = 𝒫.TR(s,a)

push!(τ, (s,a,r))

s = s′

end

return τ

end

Algorithm 11.1. A method for gen-
erating a trajectory associated with
problem 𝒫 starting in state s and
executing policy π to depth d. It
creates a vector τ containing state-
action-reward tuples.

A major challenge in arriving at accurate estimates of the policy gradient is the
fact that the variance of the trajectory rewards can be quite high. One approach to
reduce the resulting variance in the gradient estimate is to have each rollout share
the same random generator seeds.1 This approach can be helpful, for example, 1 This random seed sharing is used

in the PEGASUS algorithm. A.Y.
Ng and M. Jordan, ‘‘A Policy
Search Method for Large MDPs
and POMDPs,’’ in Conference on
Uncertainty in Artificial Intelligence
(UAI), 2000.

in cases where one rollout happens to hit a low-probability transition early on.
Other rollouts will have the same tendency due to the shared random generator,
and their rewards will tend to be biased in the same way.

Policy representations have a significant effect on the policy gradient. Exam-
ple 11.1 demonstrates the sensitivity of the policy gradient to the policy parame-
terization. Finite differences for policy optimization can perform poorly when the
parameters differ in scale.

struct FiniteDifferenceGradient

𝒫 # problem

b # initial state distribution

d # depth

m # number of samples

δ # step size

end

function gradient(M::FiniteDifferenceGradient, π, θ)

𝒫, b, d, m, δ, γ, n = M.𝒫, M.b, M.d, M.m, M.δ, M.𝒫.γ, length(θ)

Δθ(i) = [i == k ? δ : 0.0 for k in 1:n]

R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))

U(θ) = mean(R(simulate(𝒫, rand(b), s->π(θ, s), d)) for i in 1:m)

ΔU = [U(θ + Δθ(i)) - U(θ) for i in 1:n]

return ΔU ./ δ

end

Algorithm 11.2. A method for es-
timating a policy gradient using
finite differences for problem 𝒫, a
parameterized policy π(θ, s), and
policy parameterization vector θ.
Utility estimates are made from m

rollouts to depth d. The step size is
given by δ.

Consider a single-state, single-step MDP with a one-dimensional continuous
action space and a reward function R(s, a) = a. In this case, larger actions
produce higher rewards.

Suppose we have a stochastic policy πθ that samples its action according
to a uniform distribution between θ1 and θ2 for θ2 > θ1. The expected value
is:

U(θ) = E[a] =
∫ θ2

θ1

a
1

θ2 − θ1
da =

θ1 + θ2

2

The policy gradient is:

∇U(θ) = [1/2, 1/2]

The policy could be reparameterized to draw actions from a uniform
distribution between θ1 and 100θ2, for 100θ2 > θ1. Now the expected reward
is (θ1 + 100θ2)/2 and the policy gradient is [1/2, 50].

Both policies are identical from the agent’s perspective, but the two have
wildly different gradients. Finding a suitable perturbation scalar for the
second policy is much more difficult, as the parameters vary widely in scale.

Example 11.1. An example of how
policy parameterization has a sig-
nificant impact on the policy gradi-
ent.

π

11.2 Regression Gradient

Instead of estimating the gradient at θ by taking a fixed step along each coordinate
axis as done in the previous section, we can use linear regression2 to estimate the 2 Linear regression is covered in

section 8.6.gradient from the results of random perturbations from θ. These perturbations
are stored in a matrix:

∆Θ =


(∆θ(1))>

...
(∆θ(m))>

 (11.6)

More policy parameter perturbations will tend to produce better gradient esti-
mates.3 3 A recommended rule of thumb is

to use about twice as many pertur-
bations as the number of parame-
ters.

For each of these perturbations, we perform a rollout and estimate the change
in utility:4

4 This equation shows the forward
difference. Other finite difference
formulations, such as the central
difference, can also be used.

∆U =
[
U(θ+ ∆θ(1))−U(θ), . . . , U(θ+ ∆θ(m))−U(θ)

]
(11.7)

The policy gradient estimate using linear regression is then:5 5 As discussed earlier in section 8.6,
X+ denotes the pseudoinverse of
X.∇U(θ) ≈ ∆Θ+∆U (11.8)

Algorithm 11.3 provides an implementation of this approach where the pertur-
bations are drawn uniformly from a hypersphere with radius δ. Example 11.2
demonstrates this approach on a simple function.

struct RegressionGradient

𝒫 # problem

b # initial state distribution

d # depth

m # number of samples

δ # step size

end

function gradient(M::RegressionGradient, π, θ)

𝒫, b, d, m, δ, γ = M.𝒫, M.b, M.d, M.m, M.δ, M.𝒫.γ

ΔΘ = [δ.*normalize(randn(length(θ)), 2) for i = 1:m]

R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))

U(θ) = R(simulate(𝒫, rand(b), s->π(θ,s), d))

ΔU = [U(θ + Δθ) - U(θ) for Δθ in ΔΘ]

return pinv(reduce(hcat, ΔΘ)') * ΔU

end

Algorithm 11.3. A method for es-
timating a policy gradient using
finite differences for an MDP 𝒫,
a stochastic parameterized policy
π(θ, s), and policy parameteriza-
tion vector θ. Policy variation vec-
tors are generated by normalizing
normally-distributed samples and
scaling by a perturbation scalar δ.
A total of m parameter perturba-
tions are generated, and each is
evaluated in a rollout from an ini-
tial state drawn from b to depth d

and compared to the original pol-
icy parameterization.

11.2. regression gradient 229

We would like to apply the regression gradient to estimate the gradient of
a simple one-dimensional function f (x) = x2 evaluated at x0 = 2 from
m = 20 samples. To imitate the stochasticity inherent in policy evaluation, we
add noise to the function evaluations. We generate a set of disturbances ∆X
sampled fromN (0, δ2) and evaluate f (x0 +∆x)− f (x0) for each disturbance
∆x in ∆X. We can then estimate the one-dimensional gradient (or derivative)
∆X+∆F with this code:

f(x) = x^2 + 1e-2*randn()

m = 20

δ = 1e-2

ΔX = [δ.*randn() for i = 1:m]

x0 = 2.0

ΔF = [f(x0 + Δx) - f(x0) for Δx in ΔX]

pinv(ΔX) * ΔF

The samples and linear regression are shown below. The slope of the regres-
sion line is close to the exact solution of 4.

−3 −2 −1 0 1 2

×10−2

−0.10

−0.05

0.00

0.05

0.10

∆x

∆
f

∆ f = 4.216× ∆x

Example 11.2. Demonstration of
the regression gradient method on
a one-dimensional function.

230 chapter 11. policy gradient estimation

11.3 Likelihood Ratio

In some applications, we may have an analytical form of ∇πθ and can use this
to improve our estimate of ∇U(θ). The derivation of this approach uses what is
sometimes called the likelihood ratio trick or log derivative trick, which observes that
∇θ log pθ(τ) = ∇θpθ(τ)/pθ(τ). Consequently,

∇θpθ(τ) = pθ(τ)∇θ log pθ(τ) (11.9)

Recall from equation (10.2) that

U(θ) =
∫

pθ(τ)R(τ)dτ (11.10)

Hence,

∇U(θ) = ∇θ
∫

pθ(τ)R(τ)dτ (11.11)

=
∫
∇θpθ(τ)R(τ)dτ (11.12)

Applying the likelihood ratio trick, we have

∇U(θ) =
∫

pθ(τ)∇θ log pθ(τ)R(τ)dτ (11.13)

= Eτ [∇θ log pθ(τ)R(τ)] (11.14)

We can estimate the expectation above through rolling out trajectories. For each
trajectory sample τ(i), we need to compute the product∇θ log pθ(τ(i))R(τ(i)). Re-
call that R(τ(i)) is the return associated with trajectory τ(i). If we have a stochastic
policy, the gradient ∇θ log pθ(τ(i)) is

∇θ log pθ(τ) =
d

∑
k=1
∇θ log πθ(a(k) | s(k)) (11.15)

because pθ(τ) has the form:

pθ(τ) = p(s(1))
d

∏
k=1

T(s(k+1) | s(k), a(k))πθ(a(k) | s(k)) (11.16)

where s(k) and a(k) are the kth state and action, respectively, in trajectory τ. Algo-
rithm 11.4 provides an implementation where m trajectories are sampled to arrive
at a gradient estimate. Example 11.3 illustrates the process.

11.4. reward-to-go 231

If we have a deterministic policy, the gradient requires computing:6 6 Many problems have vector-
valued actions a ∈ Rn. In this case,
the ∇θπθ(s(k)) is replaced with a
Jacobian matrix whose jth column
is the gradient with respect to
the jth action component, and
the ∂

∂a(k)
log T(s(k+1) | s(k), a(k)) is

replaced with an action gradient.

∇θ log pθ(τ) = ∇θ log

[
p(s(1))

d

∏
k=1

T(s(k+1) | s(k), πθ(s(k)))

]
(11.17)

=
d

∑
k=1
∇θπθ(s(k))

∂

∂a(k)
log T(s(k+1) | s(k), a(k)) (11.18)

The equation above requires knowing transition likelihood, which is in contrast
with equation (11.15) for stochastic policies.

struct LikelihoodRatioGradient

𝒫 # problem

b # initial state distribution

d # depth

m # number of samples

∇logπ # gradient of log likelihood

end

function gradient(M::LikelihoodRatioGradient, π, θ)

𝒫, b, d, m, ∇logπ, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.𝒫.γ

πθ(s) = π(θ, s)

R(τ) = sum(r*γ^(k-1) for (k, (s,a,r)) in enumerate(τ))

∇U(τ) = sum(∇logπ(θ, a, s) for (s,a) in τ)*R(τ)

return mean(∇U(simulate(𝒫, rand(b), πθ, d)) for i in 1:m)

end

Algorithm 11.4. A method for esti-
mating a policy gradient of a pol-
icy π(s) for an MDP 𝒫 with initial
state distribution b using the likeli-
hood ratio trick. The gradient with
respect to the parameterization vec-
tor θ is estimated from m rollouts to
depth d using the log policy gradi-
ents ∇logπ.

11.4 Reward-to-Go

The likelihood ratio policy gradient method is unbiased but has high variance.
Example 11.4 reviews bias and variance. The variance generally increases sig-
nificantly with rollout depth due to the correlation between actions, states, and
rewards across time steps. The reward-to-go approach attempts to reduce the
variance in the estimate.

To derive this approach, we begin by expanding equation (11.14):

∇U(θ) = Eτ

[(
d

∑
k=1
∇θ log πθ(a(k) | s(k))

)(
d

∑
k=1

r(k)γk−1

)]
(11.19)

Let f (k) replace ∇θ log πθ(a(k) | s(k)) for convenience. We then expand:

232 chapter 11. policy gradient estimation

Consider the single-step, single-state problem from example 11.1. Suppose
we have a stochastic policy πθ that samples its action according to a Gaussian
distribution N (θ1, θ2

2) where θ2
2 is the variance.

The log policy likelihood is:

log πθ(a | s) = log

 1√
2πθ2

2

exp

(
− (a− θ1)

2

2θ2
2

)
= − (a− θ1)

2

2θ2
2
− 1

2
log
(

2πθ2
2

)
The gradient of the log policy likelihood is:

∂

∂θ1
log πθ(a | s) =

a− θ1

θ2
2

∂

∂θ2
log πθ(a | s) =

(a− θ1)
2 − θ2

2
θ3

2

Suppose we run three rollouts with θ = [0, 1], taking actions {0.5,−1, 0.7}
and receiving the same rewards (R(s, a) = a). The estimated policy gradient
is:

∇U(θ) ≈ 1
m

m

∑
i=1
∇θ log pθ(τ(i))R(τ(i))

=
1
3

([
0.5
−0.75

]
0.5 +

[
−1.0
0.0

]
(−1) +

[
0.7
−0.51

]
0.7

)
= [0.58,−0.244]

Example 11.3. Applying the likeli-
hood ratio trick to estimate a policy
gradient in a simple problem.

11.4. reward-to-go 233

When estimating a quantity of interest from a collection of simulations, we
generally want to use a scheme that has both low bias and low variance. In
this chapter, we want to estimate ∇U(θ). Generally, with more simulation
samples, we can arrive at a better estimate. Some methods can lead to bias,
where—even with infinitely many samples—it does not lead to an accurate
estimate. Sometimes methods with non-zero bias may still be attractive if
they have low variance, meaning they require fewer samples to converge.

Below are plots of the estimates from four notional methods for estimat-
ing ∇U(θ). The true value is 17.5, as indicated by the red line. For all four
methods, we ran 100 simulations 100 times. The variance decreases as the
number of samples increases. The blue regions indicate the 5 % to 95 % and
25 % to 75 % empirical quantiles of the estimates.

10

15

20

25

hi
gh

bi
as

∇
U
(θ
)

high variance low variance

100 101 102
10

15

20

25

number of samples

lo
w

bi
as

∇
U
(θ
)

100 101 102

number of samples

Example 11.4. Empirical demon-
stration of bias and variance when
estimating ∇U(θ).

∇U(θ) = Eτ

[(
d

∑
k=1

f (k)
)(

d

∑
k=1

r(k)γk−1

)]
(11.20)

= Eτ

[(
f (1) + f (2) + f (3) + · · ·+ f (d)

)(
r(1) + r(2)γ + r(3)γ2 + · · ·+ r(d)γd−1

)]
(11.21)

= Eτ


f (1)r(1) + f (1)r(2)γ + f (1)r(3)γ2 + · · ·+ f (1)r(d)γd−1

+ f (2)r(1) + f (2)r(2)γ + f (2)r(3)γ2 + · · ·+ f (2)r(d)γd−1

+ f (3)r(1) + f (3)r(2)γ + f (3)r(3)γ2 + · · ·+ f (3)r(d)γd−1

...
+ f (d)r(1) + f (d)r(2)γ + f (d)r(3)γ2 + · · ·+ f (d)r(d)γd−1

 (11.22)

The first reward, r(1), is only affected by the first action. Thus, its contribution to
the policy gradient should not depend on subsequent timesteps. We can remove
other such causality-violating terms:7 7 The term ∑d

`=k r(`)γ`−k is often
called the reward-to-go from step k.

∇U(θ) = Eτ


f (1)r(1) + f (1)r(2)γ + f (1)r(3)γ2 + · · ·+ f (1)r(d)γd−1

+ f (2)r(2)γ + f (2)r(3)γ2 + · · ·+ f (2)r(d)γd−1

+ f (3)r(3)γ2 + · · ·+ f (3)r(d)γd−1

...
+ f (d)r(d)γd−1

 (11.23)

= Eτ

[
d

∑
k=1
∇θ log πθ(a(k) | s(k))

(
d

∑
`=k

r(`)γ`−1

)]
(11.24)

= Eτ

[
d

∑
k=1
∇θ log πθ(a(k) | s(k))

(
γk−1

d

∑
`=k

r(`)γ`−k

)]
(11.25)

= Eτ

[
d

∑
k=1
∇θ log πθ(a(k) | s(k))γk−1r(k)to-go

]
(11.26)

Algorithm 11.5 provides an implementation.
Notice that the reward-to-go for a state-action pair (s, a) under a policy param-

eterized by θ is really an approximation of the state-action value from that state,
Qθ(s, a). The action value function, if known, can be used to obtain the policy
gradient:

∇U(θ) = Eτ

[
d

∑
k=1
∇θ log πθ(a(k) | s(k))γk−1Qθ

(
s(k), a(k)

)]
(11.27)

struct RewardToGoGradient

𝒫 # problem

b # initial state distribution

d # depth

m # number of samples

∇logπ # gradient of log likelihood

end

function gradient(M::RewardToGoGradient, π, θ)

𝒫, b, d, m, ∇logπ, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.𝒫.γ

πθ(s) = π(θ, s)

R(τ, j) = sum(r*γ^(k-1) for (k,(s,a,r)) in zip(j:d, τ[j:end]))

∇U(τ) = sum(∇logπ(θ, a, s)*R(τ,j) for (j, (s,a,r)) in enumerate(τ))

return mean(∇U(simulate(𝒫, rand(b), πθ, d)) for i in 1:m)

end

Algorithm 11.5. A method that
uses reward-to-go for estimating
a policy gradient of a policy π(s)

for an MDP 𝒫 with initial state dis-
tribution b. The gradient with re-
spect to the parameterization vec-
tor θ is estimated from m rollouts to
depth d using the log policy gradi-
ents ∇logπ.

11.5 Baseline Subtraction

We can further build upon the approach presented in the previous section by
subtracting a baseline value from the reward-to-go8 to reduce the variance of the 8 We could also subtract a baseline

from a state-action value.gradient estimate. This subtraction does not bias the gradient.
We now subtract a baseline rbase(s(k)):

∇U(θ) = Eτ

[
d

∑
k=1
∇θ log πθ(a(k) | s(k))γk−1

(
r(k)to-go − rbase(s(k))

)]
(11.28)

To show that baseline subtraction does not bias the gradient, we first expand:

∇U(θ) = Eτ

[
d

∑
k=1
∇θ log πθ(a(k) | s(k))γk−1r(k)to-go −

d

∑
k=1
∇θ log πθ(a(k) | s(k))γk−1rbase(s(k))

]
(11.29)

The linearity of expectation states that E[a + b] = E[a] + E[b], so it is sufficient
to prove that equation (11.29) is equivalent to equation (11.26), if for each step k,
the expected associated baseline term is 0:

Eτ

[
∇θ log πθ(a(k) | s(k))γk−1rbase(s(k))

]
= 0 (11.30)

We begin by converting the expectation into nested expectations, as illustrated
in figure 11.2:

Eτ

[
∇θ log πθ(a(k) | s(k))γk−1rbase(s(k))

]
= Eτ1:k

[
Eτk+1:d

[
∇θ log πθ(a(k) | s(k))γk−1rbase(s(k))

]]
(11.31)

Eτ [ f (τ)] Eτ1:k

[
Eτk+1:d [ f (τ)]

]

=

Figure 11.2. The expectation of
a function of trajectories sampled
from a policy can be viewed as
an expectation over a nested ex-
pectation of sub-trajectories. For a
mathematical derivation, see exer-
cise 11.3.

We continue with our derivation, using the same log derivative trick as was
used in equation (11.9):

Eτ1:k

[
Eτk+1:d

[
∇θ log πθ(a(k) | s(k))γk−1rbase(s(k))

]]
= Eτ1:k

[
γk−1rbase(s(k)) Eτk+1:d

[
∇θ log πθ(a(k) | s(k))

]]
(11.32)

= Eτ1:k

[
γk−1rbase(s(k)) Ea(k)

[
∇θ log πθ(a(k) | s(k))

]]
(11.33)

= Eτ1:k

[
γk−1rbase(s(k))

∫
∇θ log πθ(a(k) | s(k))πθ(a(k) | s(k))da(k)

]
(11.34)

= Eτ1:k

[
γk−1rbase(s(k))

∫ ∇θπθ(a(k) | s(k))
πθ(a(k) | s(k))

πθ(a(k) | s(k))da(k)
]

(11.35)

= Eτ1:k

[
γk−1rbase(s(k))∇θ

∫
πθ(a(k) | s(k))da(k)

]
(11.36)

= Eτ1:k

[
γk−1rbase(s(k))∇θ1

]
(11.37)

= Eτ1:k

[
γk−1rbase(s(k)) 0

]
(11.38)

Therefore, subtracting a term rbase(s(k)) does not bias the estimate. This derivation
assumed continuous state and action spaces. The same result applies to discrete
spaces.

We can choose a different rbase(s) for every component of the gradient, andwill
select them to minimize the variance. For simplicity, we will drop the dependence
on s and treat each baseline component as constant.9 For compactness in writing

9 Some methods approximate a
state-dependent baseline using
rbase(s(k)) = φ(s(k))>w. Selecting
appropriate baseline functions
tends to be difficult. J. Peters
and S. Schaal, ‘‘Reinforcement
Learning of Motor Skills with
Policy Gradients,’’ Neural Networks,
vol. 21, no. 4, pp. 682–697, 2008.

down the equations in our derivation, we define

`i(a, s, k) = γk−1 ∂

∂θi
log πθ(a | s) (11.39)

The variance of the ith component of our gradient estimate in equation (11.28) is

E
a,s,rto-go,k

[(
`i(a, s, k)

(
rto-go − rbase,i

))2
]
− E

a,s,rto-go,k

[
`i(a, s, k)

(
rto-go − rbase,i

)]2
(11.40)

where the expectation is over the (a, s, rto-go) tuples in our trajectory samples, and
k is each tuple’s depth.

We have just shown that the second term is zero. Hence, we can focus on
choosing rbase,i to minimize the first term by taking the derivative with respect to
the baseline and setting it to zero:

∂

∂rbase,i
E

a,s,rto-go,k

[(
`i(a, s, k)

(
rto-go − rbase,i

))2
]

=
∂

∂rbase,i

(
E

a,s,rto-go,k

[
`i(a, s, k)2r2

to-go

]
− 2 E

a,s,rto-go,k

[
`i(a, s, k)2rto-gorbase,i

]
+ r2

base,i E
a,s,k

[
`i(a, s, k)2

])
(11.41)

= −2 E
a,s,rto-go,k

[
`i(a, s, k)2rto-go

]
+ 2rbase,i E

a,s

[
`i(a, s, k)2

]
= 0 (11.42)

Solving for rbase,i yields the baseline component that minimizes the variance:

rbase,i =
Ea,s,rto-go,k

[
`i(a, s, k)2rto-go

]
Ea,s,k

[
`i(a, s, k)2

] (11.43)

It is common to use likelihood ratio policy gradient estimation with this base-
line subtraction (algorithm 11.6).10 Figure 11.3 compares the methods we dis- 10 R. J. Williams, ‘‘Simple Statistical

Gradient-Following Algorithms
for Connectionist Reinforcement
Learning,’’ Machine Learning,
vol. 8, pp. 229–256, 1992.

cussed.
Qualitatively, when considering the gradient contribution of state-action pairs,

what we really care about is the relative value of one action over another. If all
actions in a particular state produce the same high value, there is no real signal in
the gradient, and baseline subtraction can zero that out. We want to identify the
actions that produce a higher value than others, irrespective of what the mean
value is across actions.

An alternative to the action value is the advantage, A(s, a) = Q(s, a)−U(s).
Using the state value function in baseline subtraction produces the advantage.
The policy gradient using the advantage is unbiased and typically has much lower

struct BaselineSubtractionGradient

𝒫 # problem

b # initial state distribution

d # depth

m # number of samples

∇logπ # gradient of log likelihood

end

function gradient(M::BaselineSubtractionGradient, π, θ)

𝒫, b, d, m, ∇logπ, γ = M.𝒫, M.b, M.d, M.m, M.∇logπ, M.𝒫.γ

πθ(s) = π(θ, s)

ℓ(a, s, k) = ∇logπ(θ, a, s)*γ^(k-1)

R(τ, k) = sum(r*γ^(j-1) for (j,(s,a,r)) in enumerate(τ[k:end]))

numer(τ) = sum(ℓ(a,s,k).^2*R(τ,k) for (k,(s,a,r)) in enumerate(τ))

denom(τ) = sum(ℓ(a,s,k).^2 for (k,(s,a)) in enumerate(τ))

base(τ) = numer(τ) ./ denom(τ)

trajs = [simulate(𝒫, rand(b), πθ, d) for i in 1:m]

rbase = mean(base(τ) for τ in trajs)

∇U(τ) = sum(ℓ(a,s,k).*(R(τ,k).-rbase) for (k,(s,a,r)) in enumerate(τ))

return mean(∇U(τ) for τ in trajs)

end

Algorithm 11.6. Likelihood ratio
gradient estimation with reward-
to-go and baseline subtraction for
anMDP 𝒫, policy π, and initial state
distribution b. The gradient with
respect to the parameterization vec-
tor θ is estimated from m rollouts to
depth d using the log policy gradi-
ents ∇logπ.

Figure 11.3. Several policy gradient
methods used to optimize policies
for the simple regulator problem
from the same initial parameteriza-
tion. Each gradient evaluation ran
6 rollouts to depth 10. The magni-
tude of the gradient was limited to
1 and step updates were applied
with a step size 0.2. The optimal
policy parameterization is shown
in black.

variance. The gradient computation has this form:

∇U(θ) = Eτ

[
d

∑
k=1
∇θ log πθ(a(k) | s(k))γk−1 Aθ

(
s(k), a(k)

)]
(11.44)

As with the state and action value functions, the advantage function is typically
unknown. Other methods, covered in chapter 13, are needed to approximate it.

11.6 Summary

• A gradient can be estimated using finite-differences.

• Linear regression can also be used to provide more robust estimates of the
policy gradient.

• The policy gradient theorem can be used to derive a form of the policy gradient
that separates the gradient of the action likelihood, which can typically be
computed.

• The variance of the policy gradient can be significantly reduced using the
reward-to-go and baseline subtraction.

11.7 Exercises

Exercise 11.1. If we estimate the expected discounted return of a given parameterized
policy πθ defined by a n-dimensional vector of parameters θ using m rollouts, how many
total rollouts dowe need to perform to compute the policy gradient using a finite difference
approach?

Solution: In order to estimate the policy gradient using a finite difference approach, we
need to estimate the utility of the policy given the current parameter vector U(θ) as well
as given all n variations of the current parameter vector U(θ+ δe(i)) for i = 1 : n. Since
we estimate each of these using m rollouts, we need to perform a total of m(n + 1) rollouts.

Exercise 11.2. Consider policy gradients of the form:

∇θR(π) = Eτ

[
d

∑
k=1

γk−1y∇θ log πθ(a(k) | s(k))

]

Which of the following values of y result in a valid policy gradient?

(a) γ1−k ∑∞
`=1 r(`)γ`−1

240 chapter 11. policy gradient estimation

(b) ∑∞
`=k r(`)γ`−k

(c)
(

∑∞
`=k r(`)γ`−k

)
− rbase(s(k))

(d) U(s(k))

(e) Q(s(k), a(k))

(f) A(s(k), a(k))

(g) r(k) + γU(s(k+1))−U(s(k))

Solution:

(a) ∑∞
`=1 r(`) results in the total discounted reward, as

γk−1γ1−k
∞

∑
`=1

r(`)γ`−1 =
∞

∑
`=1

r(`)γ`−1

and produces a valid policy gradient as given in equation (11.19).

(b) ∑∞
`=k r(`)γ`−k is the reward-to-go, and produces a valid policy gradient as given in

equation (11.26).

(c)
(

∑∞
`=k r(`)

)
− b(s(k)) is the baseline subtracted reward-to-go, and produces a valid

policy gradient as given in equation (11.28).

(d) U(s(k)) is the state value function, and does not produce a valid policy gradient.

(e) Q(s(k), a(k)) is the state-action value function, and produces a valid policy gradient as
given in equation (11.27).

(f) A(s(k), a(k)) is the advantage function, and produces a valid policy gradient as given in
equation (11.44).

(g) r(k) + γU(s(k+1))−U(s(k)) is the temporal difference residual, and produces a valid
policy gradient because it is an unbiased approximation to the advantage function.

Exercise 11.3. Show that Eτ∼π [ f (τ)] = Eτ1:k∼π [Eτk:d∼π [ f (τ)]] for some step k.

Solution: The nested expectations can be proven by writing the expectation in integral form
and then converting back:

2021-01-07 22:28:48-08:00, revision abe81c1, draft: send comments to bugs@algorithmsbook.com



11.7. exercises 241

Eτ∼π [ f (τ)] =

=
∫

p(τ) f (τ)dτ

=
∫ (

p(s(1))
d

∏
k=1

p(s(k+1) | s(k), a(k))π(a(k) | s(k))

)
f (τ)dτ

=
∫ ∫ ∫ ∫

· · ·
∫ (

p(s(1))
d

∏
k=1

p(s(k+1) | s(k), a(k))π(a(k) | s(k))

)
f (τ)ds(d) · · ·da(2) ds(2) da(1) ds(1)

= E
τ1:k∼π

 ∫ ∫ ∫ ∫
· · ·

∫  d

∏
q=k

p(s(q+1) | s(q), a(q))π(a(q) | s(q))

 f (τ)ds(d) · · ·da(k+1) ds(k+1) da(k) ds(k)


= Eτ1:k∼π [Eτk:d∼π [ f (τ)]]

